import QtQuick 1.0
import "js/httpAccessParam.js" as HttpAccessParam

WorkerScript {
    id: _root
    source: "js/HttpAccess.js"

    property bool ready: true

    onMessage: {
        // Workerスレッドからの通信イベント
        ready = true;
        var func = HttpAccessParam.shift(messageObject.id).func;
        func(messageObject.text);
    }

    // 呼出し簡略化用
    function requestHttp( data , method , url, on_func){
        // ユニークID作成
        var id = HttpAccessParam.getId();
        // httpアクセス情報作成
        var request = {"data": data
            , "method": method
            , "url": url
            , "id": id};
        // コールバック情報保存
        var callback = {"id" : id
            , "func": on_func};
        HttpAccessParam.push(callback);
        // スレッド開始
        ready = false;
        _root.sendMessage(request);
    }
}
